Autocorrelation Coefficients—AUTO

The program AUTO (**Auto**correlation Coefficients) is a subroutine subprogram that calculates the autocorrelation coefficients of time histories when the time history data are given as equally spaced discrete values x_m (m = 0, 1, 2,⋯, N −1).

AUTO（**Auto**correlation Coefficients）

【Purpose】

To compute the autocorrelation coefficients for the given equal interval data x_m (m =0, 1, 2,⋯, N-1).

【Usage】

（1） How to connect

```
CALL AUTO (N, X, ND1, R, ND2, NFOLD)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of complex data X	Unchanged
X	R 1-D array (ND1)	Equal interval real data	Unchanged
ND1	I	Dimension size of X in calling program (ND1 .GE. N)	Unchanged
R	R 1-D array (ND2)	No need to input here	Autocorrelation coefficients
ND2	I	Dimension size of R in calling program (ND2 .GE. N/2+1)	Unchanged
NFOLD	I	No need to input here	Number of autocorrelation coefficients

(2） Necessary subroutines and function subprograms

None

(3） Remarks

i) The autocorrelation coefficients are normalized by the mean square of the data.

ii) The autocorrelation coefficients are computed at intervals equal to the time interval of the data, starting from time shift 0.

iii) The argument *NFOLD* returns N / 2 +1 if N is even, or (N +1) / 2 if N is odd.

【Calculation Method】

When the time history data are given as equally spaced discrete values x_m (m =0, 1, 2,⋯, N-1), the autocorrelation coefficients of this time history are calculated by the following equation.

$$\rho_j = \frac{\sum_{m=1}^{N} x_m x_{m+j-1}}{\sum_{m=1}^{N} x_m^{\,2}} \qquad j = 1,\ 2,\ \cdots,\ NFOLD$$

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * *                AUTO   1
C      SUBROUTINE FOR AUTOCORRELATION COEFFICIENTS                          AUTO   2
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * *                AUTO   3
C                                                                           AUTO   4
C                                  CODED BY Y.OHSAKI                        AUTO   5
C                                                                           AUTO   6
C      PURPOSE                                                              AUTO   7
C        TO COMPUTE AUTOCORRELATION COEFFICIENTS OF A SERIES OF EQUI-       AUTO   8
C        SPACED DATA                                                        AUTO   9
C                                                                           AUTO  10
C      USAGE                                                                AUTO  11
C        CALL AUTO(N,X,ND1,R,ND2,NFOLD)                                     AUTO  12
C                                                                           AUTO  13
C      DESCRIPTION OF ARGUMENTS                                             AUTO  14
C        N       - TOTAL NUMBER OF DATA                                     AUTO  15
C        X(ND1)  - EQUI-SPACED DATA                                         AUTO  16
C        ND1     - DIMENSION OF X IN CALLING PROGRAM                        AUTO  17
C        R(ND2)  - AUTOCORRELATION COEFFICIENTS                             AUTO  18
C        ND2     - DIMENSION OF R IN CALLING PROGRAM  ND2.GE.N/2+1          AUTO  19
C        NFOLD   - TOTAL NUMBER OF AUTOCORRELATION COEFFICIENTS             AUTO  20
C                                                                           AUTO  21
C      REMARKS                                                              AUTO  22
C        (1) AUTOCORRELATION COEFFICIENTS ARE NORMALIZED IN TERMS OF        AUTO  23
C            THE MEAN OF SQUARED DATA                                       AUTO  24
C        (2) AUTOCORRELATION COEFFICIENS ARE SPACED AT THE SAME INTER-      AUTO  25
C            VAL AS THE GIVEN SERIES OF DATA                                AUTO  26
C                                                                           AUTO  27
C      SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                        AUTO  28
C        NONE                                                               AUTO  29
C                                                                           AUTO  30
       SUBROUTINE AUTO(N,X,ND1,R,ND2,NFOLD)                                 AUTO  31
C                                                                           AUTO  32
       DIMENSION X(ND1),R(ND2)                                              AUTO  33
C                                                                           AUTO  34
       NFOLD=N/2+1                                                          AUTO  35
       DO 120 J=1,NFOLD                                                     AUTO  36
       RJ=0.                                                                AUTO  37
       DO 110 M=1,N                                                         AUTO  38
       MJ=M+J-1                                                             AUTO  39
       IF(MJ.GT.N) MJ=MJ-N                                                  AUTO  40
       RJ=RJ+X(M)*X(MJ)                                                     AUTO  41
```

```
  110 CONTINUE                                                    AUTO  42
      R(J)=RJ                                                     AUTO  43
  120 CONTINUE                                                    AUTO  44
      R0=R(1)                                                     AUTO  45
      DO 130 J=1,NFOLD                                            AUTO  46
      R(J)=R(J)/R0                                                AUTO  47
  130 CONTINUE                                                    AUTO  48
      RETURN                                                      AUTO  49
      END                                                         AUTO  50
```

【Example】

Calculate the autocorrelation coefficients of the time history data given by the DATA statement.

```
C
      DIMENSION DATA(16),R(9),TAU(9)
      DATA  NN/16/,DATA/5.0,32.0,38.0,-33.0,
     &                  -19.0,-10.0,1.0,-8.0,-20.0,10.0,
     &                  -1.0,4.0,11.0,-1.0,-7.0,-2.0/,DT/0.5/
C
      CALL AUTO(NN,DATA,16,R,9,NFOLD)
      DO 110 J=1,NFOLD
      TAU(J)=REAL(J-1)*DT
  110 CONTINUE
      WRITE(6,601) NN,(J-1,TAU(J),R(J),J=1,NFOLD)
      STOP
  601 FORMAT('EXAMPLE WAVE'//'-- AUTOCORRELATION COEFFICIENTS --'
     *      //T5,
     *       'TOTAL NUMBER OF DATA =',I3//T9,'J',TR4,'LAG(SEC)',TR5
     *        'R'//
     *      (I9,F10.2,F10.3) )
      END
```

Output：

```
EXAMPLE WAVE

-- AUTOCORRELATION COEFFICIENTS --

    TOTAL NUMBER OF DATA = 16

        J    LAG(SEC)      R

        0       0.00      1.000
        1       0.50      0.190
        2       1.00     -0.297
        3       1.50     -0.237
        4       2.00     -0.057
        5       2.50      0.131
        6       3.00     -0.103
        7       3.50     -0.097
        8       4.00     -0.058
```